

\$8.95

37 TIMEX 1000 SINCLAIR ZX-81 Programs For Home, School, Office

A handy collection of ready-to-run software for businessmen, teachers, students and hobbyists.



by Edward Page

37 **TIMEX 1000**
SINCLAIR ZX-81
Programs For
Home, School, Office

37 **TIMEX 1000** **SINCLAIR ZX-81** **Programs For** **Home, School, Office**

by Edward Page

ARCsoft Publishers
WOODSBORO, MARYLAND

**FIRST EDITION
SECOND PRINTING**

© 1983 by ARCsoft Publishers, P.O. Box 132, Woodsboro, MD 21798 USA

Printed in the United States of America

Reproduction or publication of the contents of this book, in any manner, without express permission of the publisher, is prohibited. No liability is assumed with respect to the use of the information herein.

Library of Congress (LC) number: 82-18411

Trademark credits and software copyrights:

**Sinclair ZX-81 is a trademark of Sinclair Research Ltd.
Timex-Sinclair 1000 is a trademark of Timex Computer Corp.
Programming advice and applications software in this book
are Copyright 1983 by ARCsoft Publishers**

ISBN 0-86668-021-7

Preface

As of this writing, more TIMEX/Sinclair 1000 and Sinclair ZX-81 computers have been sold than any other personal computer in the world. Surely, that makes the Sinclair machine the world's most popular computer!

The extremely lightweight design and high portability, with a built-in powerful BASIC language, place the T/S 1000 and ZX-81 at the forefront of today's exciting computer scene.

Not a toy, the Sinclair hardware and its versatile system software make it a highly useful tool in the business environment and classroom as well as for practical jobs around the home.

In fact, the system software is so flexible that the need for this book was revealed. There are so many computer tasks which can be accomplished with the T/S 1000 and ZX-81 that the first 37 must be presented here.

Software programs make a computer do work. This book, for newcomers, beginners, novices, student programmers and even old-timers wanting new ideas, is designed to be a source of useful programs to make your personal computer do work.

This book is a companion volume to *101 TIMEX 1000/Sinclair ZX-81 Programming Tips & Tricks*.

—Edward Page

Table of Contents

11 Introduction

Programs for the home

- 19 Super Slot-O**
- 22 Monthly Loan Payment**
- 23 Number of Days In A Month**
- 25 High/Low Bowling Score**
- 26 Random Number Quality Checker**
- 28 Savings Quickie**
- 30 Draw Straws**
- 31 Funny Similes**
- 33 Car Payments**
- 35 Shopper's Friend**

Programs for the classroom

- 41 Foreign Capitals**
- 43 Areas**
- 46 Photography: Flash Exposure**
- 48 Photography: Close Ups**
- 50 Math Flasher**
- 52 Advanced Math Flasher**

- 56 Exam Score Sorting
- 58 Astronomy
- 61 State Geographic Centers
- 64 Volumes
- 67 Event Timer
- 69 Class Roll Sorter

Programs for the business person

- 73 Marketing Performance
- 74 Daily Codes
- 75 Invoice Computer
- 77 Hourly Wages
- 78 Ad Campaign Profit
- 80 Media Money Massage
- 82 Salesman's Commission
- 84 Unit Price
- 86 Executive Decision Maker
- 87 Cash Receipts Comparer
- 88 Mark Up

Appendix

- 91 BASIC Words Used in the TIMEX/Sinclair
- 94 Error Messages/Report Codes

Introduction

Introduction

There is a need for practical, useful software for the new generation of popular personal computers. This book is a collection of 37 useful programs you can type easily into your TIMEX/Sinclair 1000 and Sinclair ZX-81 computer, and run them just as you find them here.

The T/S 1000 and ZX-81 computers are one and the same, except the T/S 1000 has more memory capacity than the ZX-81. Both computers can be augmented with plug-in extra memory.

All 37 programs in this book have been thoroughly tested and run on the T/S 1000 and the ZX-81. All will run in the limited memory space available on the ZX-81. Thus, all will run on the T/S 1000, also. All of the programs in this book will run on either BASIC computer.

The T/S 1000 and the ZX-81 are powerful and versatile and flexible—but what can they do? Once you own the hardware, you need real programs designed to do real work. The aim of this book is to provide 37 all-new ready-

to-run pieces of complete software you can put to immediate use.

The 37 programs in this book can stand alone, or be used as parts of larger sets of instructions you might write. These are designed to be typed in, just as you find them here, with no other programming needed.

These programs are useful in themselves. They also make good starting points for further development as you learn more and more about how to program your own computer. Use these practical programs now and then expand them later as your imagination expands.

This book can be used by novices, student programmers, newcomers, beginners and even old-timers who need new ideas. We make the assumption you know how to set up your T/S 1000 or ZX-81 or MicroAce computer for use. You know how to connect the cable between TV switchbox and computer. You know how to plug the power supply into a wall socket and attach the power cord to the computer. You know how to type on the flat keyboard, locate the SHIFT key, and obtain the various "levels" of functions on the multiple-function keys.

If you don't have these areas of skill down pat, yet, check your owner's manual. The Sinclair ZX-81 manual is entitled *ZX-81 BASIC Programming*. The similar publication for the T/S 1000 is the *Timex User Manual*. Please review these books before starting to use the programs in this book.

You do not have to be a programmer to use the software in this book. Just type in the programs, as you find them here, and run them. They will work!

Learning

You will learn a lot from this book just by typing in the programs before running them. Please be careful to type each correctly, exactly as you find it here. If a syntax error report code, or other report code, appears, go back and make sure your type line matches what we have printed here. It's easy to make a typo!

Our program lines are expected to be exactly correct as they are printed here. All have been tested on the T/S 1000 and ZX-81. If, after typing in a line as we have it, you get an error report code, check the handy list of error

messages in the Appendix at the rear of this book. You also will find, in the Appendix, a handy list of BASIC words as used in the TIMEX/Sinclair 1000 and Sinclair ZX-81.

If you get an error message report code, most likely you will find you have made a typographical error in typing the program into your computer. However, should you find a typo which we have made in this book, please let the author know. Send a postcard or letter to the author in care of *ARCsoft Publishers*, P.O. Box 132, Woodsboro, MD 21798 USA. The author will appreciate being able to make necessary corrections to future editions.

Office, school & home

This book has three sections and an Appendix. The first section includes programs thought of as useful around the home. The second section includes programs for students and teachers and the third section is business oriented.

Naturally, these sections are not rigidly divided. You probably will find something in the business or classroom sections applicable to home use. And you probably will take several of the home programs to your office or classroom.

Try them all. They're great fun to run. And they are designed especially to be short so you won't have to spend hours typing one program.

Report codes

Sinclair calls its error messages *report codes*. We often refer to them as error messages because that's how most programmers know them. Report codes and error messages mean the same thing in this book.

Many of the programs in this book will continue to run until you press the BREAK key to command them off. You may stop any run, at any time, by pressing the BREAK key at the lower right-hand side of the keyboard.

The author would like to have your suggestions for changes in future editions of this work, or for other books in this series for the TIMEX/Sinclair 1000 and Sinclair ZX-81. He may be addressed in care of *ARCsoft Publishers*.

RETURN vs. ENTER

Programmers generally mix the two words, RETURN and ENTER, and use them to mean the same thing. In either case, when you see RETURN or ENTER in this book, we mean the ENTER key on the right-hand side of the keyboard.

These programs will run on any machine programmable in BASIC. However, to run on hardware other than the T/S 1000 or ZX-81, you will have to make slight modifications to program lines. Graphics commands, especially, will differ on non-Sinclair computers.

Refer to the owner's manual which came with your non-Sinclair computer and compare its version of BASIC with the Sinclair BASIC. A list of Sinclair BASIC words can be found in the Appendix at the rear of this book.

Also, if you use a computer other than a T/S 1000 or ZX-81, such things as line numbering, logical tests, multiplication symbols, print statements and other instructions may differ.

REMARKS

As you read through the 37 programs in this book, you will find new uses of REM, or remarks in software. The author's training in writing BASIC-language computer programs included an emphasis on brevity and saving memory space. A sharp editing pencil was in order—and still is!

With only one or two kilobytes (1k or 2k) of memory available in the ZX-81 or T/S 1000, you need to create lean programs. There is no room for fat—and this is good! Programmers who have learned on such systems make better professionals because they know how to write efficient "code."

REMARKS and explanations in software are out. Honing, fine-tuning and waste trimming are in. Use of coding-form program worksheets, such as the TIMEX 1000/Sinclair ZX-81 BASIC Coding Form, published by ARCsoft Publishers, is encouraged. The objective always is to make the most efficient use of available memory.

And be sure to remember: even though they may be headed toward the very same goal, no two programmers will write exactly the same list of BASIC instructions, or

program lines, from scratch. As you load these 37 programs into your computer, one at a time, you may make modifications to suit your personal needs and interests. For instance, exact wordings of PRINT statements can be changed. Or, two or more programs can be combined into one grand scheme. Your applications may vary.

If you want to load more than one of these programs into your T/S 1000 or ZX-81 at the same time, be sure to use different sets of line numbers for different programs. Two programs cannot occupy the same set of line numbers at the same time.

Good programming!

Programs for the home

Super Slot-0

Oh, those evil slot machines! They're just popping up everywhere. Even inside myTIMEXComputer.

Like any good slot machine, when you pull the handle it displays some objects. If you get no two alike, you lose. If you get two alike among the three objects, you win small. If all three are the same, you win big.

One difference in our Slot-O game, the display is entirely at random. No one pushes a secret button under the table to make certain items pop up.

The "handle" is the keyboard. To "pull" it, just press any key. The display objects are small boxes. If you get two alike, the computer tells you that you have won \$5. If you get three alike, the computer tells you that you have won \$10.

Get out your funny-money from that old Monopoly game, gather up your friends, and let's have some fun.

Let's start our Fun-N-Games with instruction on how to simulate a slot machine on the screen of your TIMEX!

As with all the programs in this book, simply type this one in and RUN it. The TIMEX T/S 1000 will display, on your video screen, the name of this program and some simple instructions.

To simulate pulling the slot machine's lever arm, press any key on the keyboard.

By the way, be very careful in typing in the program.

As you key in programs throughout this book, be sure you include all blank spaces where called for.

Program Listing

```
10 PRINT "PLAY SUPER SLOT-O"  
20 FOR L=1 TO 17  
30 PRINT "*";  
40 NEXT L  
50 PRINT  
60 GOSUB 800  
70 GOSUB 900
```

```

80 RAND
90 IF INKEY$="" THEN GOTO 90
100 CLS
110 GOSUB 500
120 LET A=X
130 GOSUB 500
140 LET B=X
150 GOSUB 500
160 LET C=X
170 IF A=B THEN LET P=1
180 IF B=C THEN LET Q=1
190 IF C=A THEN LET R=1
200 PRINT
210 PRINT CHR$ A;" ";CHR$ B;" ";CHR$ C
220 IF P+Q+R=1 THEN LET S=1
230 IF P+Q+R>1 THEN LET T=1
240 IF P+Q+R<1 THEN LET U=1
300 PRINT
310 PAUSE 60
320 IF S THEN PRINT "TWO OF A KIND"
330 IF S THEN PRINT "YOU WIN $5"
340 IF T THEN PRINT "THREE OF A KIND"
350 IF T THEN PRINT "YOU WIN $10"
360 IF U THEN PRINT "SORRY, THERE IS
      NO MATCH"
370 IF U THEN PRINT "YOU DO NOT WIN ANY
      MONEY"
380 PRINT
390 GOTO 20
500 LET X=2*RND
510 IF X<1 THEN GOTO 600
520 GOTO 700
600 LET X=INT(8*RND)
610 IF X<2 THEN GOTO 600
620 RETURN
700 LET X=INT(133*RND)
710 IF X<128 THEN GOTO 700
720 RETURN
800 PRINT "PRESS ANY KEY TO"
810 PRINT "PULL THE SLOT MACHINE HANDLE"
820 RETURN

```

```
900 LET P=Ø
910 LET Q=Ø
920 LET R=Ø
930 LET S=Ø
940 LET T=Ø
950 LET U=Ø
960 RETURN
```

Monthly Loan Payment

Here's a fast computation of the monthly payment on a loan. The amount borrowed, the principle, is stored in memory location P. I is the annual interest rate and N is the number of payments. I is converted to a monthly interest rate and then to a decimal in line 50.

Program Listing

```
10 CLS
15 LET M=0
20 PRINT"AMT BORROWED = $";
25 INPUT P
30 PRINT P
35 PRINT "ANNUAL INTEREST = ";
40 INPUT I
45 PRINT I
50 LET I=(0.01*I)/12
55 PRINT "NO. OF PAYMENTS = ";
60 INPUT N
65 PRINT N
70 LET M=P*I/(1-((1+I)**(-N)))
75 PRINT "MO. PAYMENT = $";M
80 IF INKEY$=""THEN GOTO 80
85 GOTO 10
```

Sample Run

```
AMT BORROWED = $
1500   ENTER
ANNUAL INTEREST % =
18   ENTER
NO. OF PAYMENTS =
24   ENTER

MO. PAYMENT = $74.886152
```

Number Of Days In A Month

Here's how to use the computer's ability to search for data, to create a Q&A.

How many days are there in a month? It's a tough question for grade schoolers and some extra study may be in order. Use this program.

The program has the computer present the number of a month and then ask how many days are in that particular month.

If you enter a correct answer, the computer will say so. If you enter an incorrect answer, the computer will say your answer was wrong and tell the correct answer.

The program runs on forever until you press the BREAK key.

Program Listing

```
10 DIM D(12)
20 RAND
100 LET D(1)=31
110 LET D(2)=28
120 LET D(3)=31
130 LET D(4)=30
140 LET D(5)=31
150 LET D(6)=30
160 LET D(7)=31
170 LET D(8)=31
180 LET D(9)=30
190 LET D(10)=31
200 LET D(11)=30
210 LET D(12)=31
300 LET R=INT(13*RND)
310 IF R<1 THEN GOTO 300
400 PRINT "HOW MANY DAYS IN MONTH ";R
410 INPUT D
420 CLS
430 IF D=D(R) THEN GOTO 600
500 PRINT "WRONG"
510 GOTO 610
```

```
600 PRINT "CORRECT"  
610 PRINT "MONTH ";R;" HAS ";D(R);" DAYS"  
620 PRINT  
630 GOTO 10
```


High/Low Bowling Score

Suppose you bowl with a group of friends, each with a different score or set of scores? This program accepts their scores and sorts out the highest and the lowest bowling scores.

Here's how it works: at line 20 the program is dimensioned to hold data on 10 persons. Lines 30 to 70 take in the info on each person.

Naturally, this kind of sorting could be applied to any game with ranges of scores among different players.

Program Listing

```
10 CLS
20 DIM M(10)
30 FOR L=1 TO 10
40 PRINT "SCORE: ";
50 INPUT M(L)
60 PRINT M(L)
70 NEXT L
80 PRINT
90 PRINT "SORTING"
100 LET T=0
110 FOR L=1 TO 9
120 IF M(L)<=M(L+1) THEN GOTO 170
130 LET E=M(L)
140 LET M(L)=M(L+1)
150 LET M(L+1)=E
160 LET T=1
170 NEXT L
180 IF T=1 THEN GOTO 100
190 CLS
200 PRINT "SCORES IN ORDER:"
210 FOR L=1 TO 10
220 PRINT M(L)
230 NEXT L
```

Random Number Quality Checker

Ever wonder just how unintentional, haphazard, or unrelated your random numbers are? This program reinforces your confidence in the pseudorandom number generator in the TIMEX Computer.

It causes the machine to generate 100 numbers between zero and 100 and reports how many are above 49 and how many are below 50.

Program Listing

```
10 LET N=0
15 LET X=0
20 LET Y=0
25 FOR L=1 TO 100
30 LET X=INT(100*RND)
35 IF X<50 THEN LET Y=Y+1
40 IF X>49 THEN LET N=N+1
45 NEXT L
50 PRINT Y;" BELOW 50"
55 PRINT N;" ABOVE 49"
60 PRINT
65 GOTO 10
```

Ever wonder how "random" are the numbers generated by the random-number generator in your TIMEX when you use the RND instruction? Try this program.

It generates 100 random numbers in a range from zero to four and counts how many there are of each number between zero and four.

By the way, while it is doing that it will display the message "counting" so you can tell it is working.

At the end of its run, the TIMEX prints a neat chart, on the video display, of results.

Program Listing

```
10 CLEAR
20 RAND
30 LET N=Ø
40 LET T=Ø
50 DIM A(5)
60 FOR L=1 TO 100
70 PRINT "COUNTING"
80 LET N=INT(5*RND)
90 IF N=Ø THEN LET A(1)=A(1)+1
100 IF N=1 THEN LET A(2)=A(2)+1
110 IF N=2 THEN LET A(3)=A(3)+1
120 IF N=3 THEN LET A(4)=A(4)+1
130 IF N=4 THEN LET A(5)=A(5)+1
140 CLS
150 NEXT L
160 FOR L=1 TO 5
170 LET T=T+A(L)
180 PRINT L-1,A(L)
190 NEXT L
200 PRINT "TOTAL",T
```

Savings Quickie

Want a quick idea of how much your savings account will grow over the years? This program is fast to load and speedy to run.

The computer will ask for initial savings balance, annual interest percentage rate, and number of years. In return, it computes compound interest and displays the savings balance at the end of each year in a handy list.

Program Listing

```
10 CLS
20 LET Z=0
30 PRINT "PRESENT BALANCE: ";
40 INPUT B
50 PRINT B
60 PRINT "INTEREST RATE: ";
70 INPUT I
80 PRINT I
90 PRINT "NUMBER OF YEARS: ";
100 INPUT Y
110 PRINT Y
120 FOR L=1 TO Y
130 LET Z=Z+I*(Z+B)/100
140 PRINT L,Z+B
150 NEXT L
160 IF INKEY$="" THEN GOTO 160
170 GOTO 10
```

Sample Run

RUN ENTER

```
PRESENT BALANCE:
652      ENTER
INTEREST RATE:
8        ENTER
NUMBER OF YEARS:
11       ENTER
```

1	704.16
2	760.4928
3	821.33223
4	887.0388
5	958.00191
6	1034.6421
7	1117.41.34
8	1206.8065
9	1303.351
10	1407.6191
11	1520.2286

Draw Straws

Drawing straws is one of man's oldest decision makers. Several straws are broken off to various lengths. The length of all straws is concealed and each person draws a straw. The person drawing the shortest straw "wins." That is, he is selected by the luck of the draw.

Now your TIMEX Computer can provide a fast and easy drawing where no straws are available. It does all the work for you by electronically assigning straw lengths. You can decide beforehand whether longest straw or shortest straw wins.

The program is set up for eight persons to draw. The computer first asks their names. Then it secretly makes a random number assignment to each. That number feeds the computer graphics which display the straw lengths.

Program Listing

```
10 RAND
20 DIM S(8)
30 DIM N$(8,10)
100 FOR P=1 TO 8
110 PRINT "NAME OF PLAYER"
120 PRINT "NUMBER OF ";P;" ?"
130 INPUT N$(P)
140 LET S(P)=20*RND
150 CLS
160 NEXT P
200 FOR P=1 TO 8
210 PRINT N$(P);
220 FOR L=0 TO S(P)
230 PRINT CHR$ 128;
240 NEXT L
250 PRINT
260 PRINT
270 NEXT P
300 IF INKEY$="" THEN GOTO 300
310 CLS
320 GOTO 10
```


Funny similes

Give these newfangled gadgets an inch and they'll take a mile. In the case of the computer, give it some tacky retorts and it will spew out an endless string of dumb remarks.

The fun is in having the computer randomly select various words and combine them to make silly sayings. In this program, the stupidities are stored via lines 100 to 290.

The random number is used to match the words into similes.

Program Listing

```
10 RAND
20 DIM M$(10,5)
30 DIM N$(10,6)
100 LET M$(1)="FAT"
110 LET M$(2)="DIRTY"
120 LET M$(3)="BAD"
130 LET M$(4)="SAD"
140 LET M$(5)="GREEN"
150 LET M$(6)="UGLY"
160 LET M$(7)="DULL"
170 LET M$(8)="TACKY"
180 LET M$(9)="WEAK"
190 LET M$(10)="DUMB"
200 LET N$(1)="TREE"
210 LET N$(2)="PIG"
220 LET N$(3)="TURKEY"
230 LET N$(4)="DOG"
240 LET N$(5)="ROOKIE"
250 LET N$(6)="SIN"
260 LET N$(7)="PLUG"
270 LET N$(8)="BULL"
280 LET N$(9)="WORM"
290 LET N$(10)="OX"
300 PRINT "WHOM ARE WE DESCRIBING?"
310 INPUT B$
320 CLS
```

```
330 LET T=INT(11*RND)
340 IF T<1 THEN GOTO 330
350 PRINT B$;" IS ";M$(T);" AS A ";N$(T)
360 PRINT
370 GOTO 330
```

Car Payments

Shopping for a new car? Use your TIMEX T/S 1000 Computer to compute quickly the potential monthly car payment on various models.

Imagine you want an \$8000 car and are prepared to put up \$1000 against the purchase. You want to arrange to finance the car for 36 months. You know the current annual interest rate on car loans is 15 percent.

Key in those few numbers and the computer instantly tells you the car payment will be \$242.66 per month.

Program Listing

```
10 LET P=0
20 PRINT "CAR PAYMENT"
30 FOR L=1 TO 11
40 PRINT "**";
50 NEXT L
60 PRINT
100 PRINT "PURCHASE PRICE ?"
110 INPUT T
120 CLS
130 PRINT "DOWN PAYMENT ?"
140 INPUT R
150 CLS
160 PRINT "NUMBER OF MONTHS ?"
170 INPUT N
180 CLS
190 PRINT "ANNUAL INTEREST ?"
200 INPUT I
210 LET I=(0.01*I)/12
220 CLS
230 LET P=(T-R)*I/(1-1/(1+I)**N)
240 LET P=INT(100*P+.5)/100
250 PRINT "PAYMENT WILL BE"
260 PRINT "$";P;" A MONTH"
```

Sample Run

CAR PAYMENT

PURCHASE PRICE \$?

8000

DOWN PAYMENT \$?

1000

NUMBER OF MONTHS ?

36

MONTHLY INTEREST ?

15/12

PAYMENT: \$ 242.66

Shopper's Friend

Take your computer along the next time you go shopping!

This program finds the computer asking for certain information and then telling you which product brand name is the best buy.

The computer will ask for the brand name of a product, the quantity in the product package, and the price of the package. Then it will ask for the name, quantity and price for a second product.

After digesting all this information, it will tell you the brand name of the best-buy product and show you the unit prices for both brand names so you can agree with the computer's judgment.

For example, suppose you were looking at corn flakes in boxes, one by Post and one by Kellogg. Suppose the Post box contained 24 ounces of flakes and was priced on the grocery shelf at \$1.98 while the Kellogg box held 18 ounces and was priced at \$1.59. Which would be the better buy based on unit price per ounce of flakes?

Run the data through your TIMEX and you'll find it computes the Post corn flakes to be the best buy with a unit price of 8¢ vs. the Kellogg unit price of 9¢.

After completing the computation, the computer asks for a yes or no answer. Do you have more to be computed?

By the way, if the unit prices turn out to be equal, the computer will say they are equal. If you want no further computations, it will say thanks and quit.

Program Listing

```
10 PRINT "SHOPPERS FRIEND"
20 FOR L=1 TO 15
30 PRINT "**";
40 NEXT L
50 PRINT
100 PRINT "FIRST BRAND NAME:"
110 INPUT X$
120 PRINT X$
```

```

130 PRINT "QUANTITY: ";
140 INPUT M
150 PRINT M
160 PRINT "PRICE: $";
170 INPUT N
180 PRINT N
200 PRINT "SECOND BRAND NAME:"
210 INPUT Y$
220 PRINT Y$
230 PRINT "QUANTITY: ";
240 INPUT Q
250 PRINT Q
260 PRINT "PRICE: ";
270 INPUT R
280 PRINT R
300 IF N/M=R/Q THEN GOTO 600
310 IF N/M<R/Q THEN GOTO 400
320 PRINT Y$;" IS BEST BUY"
330 GOTO 500
400 PRINT X$;" IS BEST BUY"
500 PRINT X$;" UNIT = $";N/M
510 PRINT Y$;" UNIT = $";R/Q
520 PRINT
530 STOP
600 PRINT X$;" = ";Y$
610 GOTO 500

```

Sample Run

SHOPPERS FRIEND

FIRST BRAND NAME ?

POST

QUANTITY ?

24

PRICE ?

1.98

SECOND BRAND NAME ?
KELLOGS
QUANTITY ?
18
PRICE ?
1.59

POST IS BEST BUY
POST UNIT=\$0.08
KELLOGS UNIT=\$0.09

FIRST BRAND NAME ?
BLACKS
QUANTITY ?
100
PRICE ?
25
SECOND BRAND NAME ?
WHITES
QUANTITY ?
50
PRICE ?
12.5

BLACKS = WHITES
BLACKS UNIT=\$0.25
WHITES UNIT=\$0.25

Programs for the classroom

Foreign Capitals

Here's a learning quiz we'll bet you haven't seen anywhere else. This program tests your knowledge of foreign countries. The more you play, the more you learn!

You must tell the computer the correct name of the capital of the country it presents. And you must spell the name of that city correctly.

What is the capital of Egypt, Poland, Turkey, New Zealand, Bolivia or Afghanistan? It can be very tough!

Want to change to different countries? Change lines 100 to 190.

Program Listing

```
10 RAND
20 DIM C$(5,11)
30 DIM T$(5,6)
100 LET C$(1)="AFGHANISTAN"
110 LET T$(1)="KABUL"
120 LET C$(2)="BOLIVIA"
130 LET T$(2)="LA PAZ"
140 LET C$(3)="EGYPT"
150 LET T$(3)="CAIRO"
160 LET C$(4)="POLAND"
170 LET T$(4)="WARSAW"
180 LET C$(5)="TURKEY"
190 LET T$(5)="ANKARA"
200 LET R=INT(6*RND)
210 IF R<1 THEN GOTO 200
300 PRINT "COUNTRY: ";C$(R)
310 PRINT "CAPITAL?"
320 INPUT A$
330 IF A$="" THEN GOTO 320
340 LET L=LEN A$
350 CLS
360 IF A$=T$(R,1 TO L) THEN GOTO 500
400 PRINT "WRONG"
410 GOTO 510
500 PRINT "CORRECT"
510 PRINT "COUNTRY: ";C$(R)
```

```
520 PRINT "CAPITAL: ";T$(R)
530 PRINT
540 GOTO 10
```

Areas

Circle. Ellipse. Parabola. Sphere. Square. Rectangle. Triangle. Name your shape. This program will compute its area. Surface area in the case of the sphere. Answer the computer's questions and it will give you the answer you need, in square units of measure. If you use inches, the answer will be in square inches. Put in yards and get square yards. Meters, get square meters. Please don't mix units in any one computation.

Program Listing

```
100 PRINT "SHAPE?"
110 INPUT X$
120 CLS
130 PRINT X$
140 IF X$="CIRCLE" OR X$="SPHERE" THEN
    GOTO 200
150 IF X$="PARABOLA" OR X$="TRIANGLE"
    THEN GOTO 300
160 IF X$="SQUARE" OR X$="RECTANGLE"
    THEN GOTO 400
170 IF X$="ELLIPSE" THEN GOTO 500
180 GOTO 100
200 PRINT "RADIUS ";
210 INPUT R
220 PRINT R
230 LET A=PI*(R**2)
240 IF X$(1)="S" THEN LET A=4*A
250 GOTO 600
300 PRINT "BASE ";
310 INPUT B
320 PRINT B
330 PRINT "HEIGHTH ";
340 INPUT H
350 PRINT H
360 IF X$(1)="P" THEN LET A=B*H*2/3
370 IF X$(1)="T" THEN LET A=B*H*.5
380 GOTO 600
```

```

400 PRINT "WIDTH ";
410 INPUT W
420 PRINT W
430 PRINT "LENGTH ";
440 INPUT L
450 PRINT L
460 LET A=L*W
470 GOTO 600
500 PRINT "MAJOR AXIS ";
510 INPUT J
520 PRINT J
530 PRINT "MINOR AXIS ";
540 INPUT N
550 PRINT N
560 LET A=.7854*J*N
600 PRINT "AREA ";A
610 PRINT
620 GOTO 100

```

Sample Run

```

SHAPE
TRIANGLE  ENTER
BASE
55  ENTER
HEIGHTH
22  ENTER
ARFA    605

```

```

SHAPE
SPHERE  ENTER
RADIUS
49  ENTER
AREA    30171.85

```

SHAPE
ELLIPSE ENTER
MAJOR AXIS
19 ENTER
MINOR AXIS
14 ENTER
AREA 208.9164

SHAPE
PARABOLA ENTER
BASE
71 ENTER
HEIGHT
99 ENTER
AREA 4686

Photography: Flash Exposure

Use your computer to help take better pictures!

The most important factor in pictures shot with flash is the distance from your flash to the subject. Subjects which are close to you will receive a lot of light while subjects farther away will receive less light.

Check your data sheet for the film you are using. Look for the film guide number. Next, make an estimate of the distance in feet from the flash to your subject.

This program determines the proper *f*/stop setting for your camera. By the way, if the computer tells you to use an *f*/stop setting between two *f*/numbers available on your camera, set your lens opening at the nearest *f*/number or halfway between the two, whichever is closest.

For example, suppose your film has a guide number of 80 and you estimate the flash-to-subject distance at 10 feet. Use *f*/8 on your lens.

Program Listing

```
10 PRINT "FLASH EXPOSURE"
20 PRINT "*****"
30 PRINT "FILM GUIDE NUMBER: ";
40 INPUT G
50 PRINT G
60 PRINT "FLASH-SUBJECT DISTANCE: ";
70 INPUT D
80 PRINT D; " FEET"
90 LET F=G/D
100 PRINT "SHOOT AT F/";F
110 IF INKEY$ ="" THEN GOTO 110
120 CLS
130 GOTO 10
```


Sample Run

FLASH EXPOSURE

FILM GUIDE NUMBER:

80 ENTER

FLASH-SUBJECT DISTANCE:

10 ENTER

SHOOT AT F/8

Photography: Close Ups

For copying and other close-up work with your camera, you extend the camera lens by using bellows or extension tubes. In doing that, you must allow for an effective increase in the normal f /number or your picture will be underexposed.

You make such an exposure compensation whenever the subject distance is less than eight times the focal length of your lens.

This program provides a convenient means of determining the effective f /number. For example, if the focal length of your camera is 50mm and the lens-to-film distance (focal length plus extension from infinity position) is 100mm, and the normal f /stop would be 22, the corrected stop would be $f/11$.

Or, if you are using a 25mm lens, with 50mm lens-to-film distance, a normal f /stop of 8 should be corrected to $f/4$. Be sure to keep both focal length and distance in either mm or inches. Don't mix apples and oranges.

Program Listing

```
10 PRINT "CLOSE UPS"
20 PRINT "*****"
30 PRINT "NORMAL F/ NUMBER: ";
40 INPUT F
50 PRINT F
60 PRINT "LENS-FILM DISTANCE: ";
70 INPUT D
80 IF D=0 THEN GOTO 70
90 PRINT D
100 PRINT "LENS FOCAL LENGTH: ";
110 INPUT L
120 PRINT L
130 LET N=F*L/D
140 PRINT "EFFECTIVE F/NUMBER IS F/" ; N
150 FOR Z=1 TO 8
160 PRINT
170 NEXT Z
180 PRINT "FOR MORE, PRESS ANY KEY"
```

```
190 IF INKEY$="" THEN GOTO 190
200 CLS
210 GOTO 10
```

Sample Run

CLOSE UPS

NORMAL F/ NUMBER:

22 ENTER

LENS-FILM DISTANCE:

100 ENTER

LENS FOCAL LENGTH:

50 ENTER

EFFECTIVE F/NUMBER IS F/11

FOR MORE, PRESS ANY KEY

Math Flasher

Flash cards for memorizing and practicing math problems have been around since Grandpa was a boy. They used to be printed cards which you manually displayed to the testee. Now, the computer does the job and all you have to do is sit by and watch.

With this program, you select addition, subtraction, multiplication or division. The computer randomly selects a pair of numbers and creates the necessary math problem.

We assume it is desirable to *not* have negative numbers as results of subtraction. We want subtraction problems which result in answers of zero, one, two and higher. None below zero such as -1, -2, -3 or lower. The program will present only pairs of numbers which result in the desired values.

Also, in division, we want whole-number answers like 2, 11 or 26. Not 1.81, 9.7, or 21.334. The program controls for the desired answers.

Program Listing

```
10 PRINT "ADDITION"
20 FOR L=1 TO 8
30 PRINT "*";
40 NEXT L
50 PRINT
60 PRINT
100 LET P=INT(10*RND)
110 LET Q=INT(10*RND)
200 PRINT P;" PLUS ";Q
210 INPUT R
220 CLS
300 IF R=P+Q THEN GOTO 400
310 PRINT "WRONG"
320 GOTO 410
400 PRINT "CORRECT"
410 PRINT P;" PLUS ";Q;" = ";P+Q
500 PRINT
510 PRINT
520 GOTO 10
```

Program Listing

```
10 PRINT "SUBTRACTION"
20 FOR L=1 TO 11
30 PRINT "**";
40 NEXT L
50 PRINT
60 PRINT
100 LET P=INT(10*RND)
110 LET Q=INT(10*RND)
120 IF P-Q<0 THEN GOTO 100
200 PRINT "SUBTRACT ";Q;" FROM ";P
210 INPUT R
220 CLS
300 IF R=P-Q THEN GOTO 400
310 PRINT "WRONG"
320 GOTO 410
400 PRINT "CORRECT"
410 PRINT P;" MINUS ";Q;" = ";P-Q
500 PRINT
510 PRINT
520 GOTO 10
```

Program Listing

```
10 PRINT "MULTIPLICATION"
20 FOR L=1 TO 14
30 PRINT "**";
40 NEXT L
50 PRINT
60 PRINT
100 LET P=INT(10*RND)
110 LET Q=INT(10*RND)
200 PRINT "MULTIPLY ";P;" TIMES ";Q
210 INPUT R
220 CLS
300 IF R=P*Q THEN GOTO 400
310 PRINT "WRONG"
320 GOTO 410
400 PRINT "CORRECT"
410 PRINT P;" TIMES ";Q;" = ";P*Q
500 PRINT
```

```
510 PRINT
520 GOTO 10
```

Program Listing

```
10 PRINT "DIVISION"
20 FOR L=1 TO 8
30 PRINT "**";
40 NEXT L
50 PRINT
60 PRINT
100 LET P=INT(10*RND)
110 LET Q=INT(10*RND)
120 IF Q<1 THEN GOTO 110
130 IF P/Q <> INT(P/Q) THEN GOTO 100
200 PRINT "DIVIDE ";P;" BY ";Q
210 INPUT R
220 CLS
300 IF R=P/Q THEN GOTO 400
310 PRINT "WRONG"
320 GOTO 410
400 PRINT "CORRECT"
410 PRINT P;" DIVIDED BY ";Q;" = ";P/Q
500 PRINT
510 PRINT
520 GOTO 10
```

Advanced Math Flasher

Similar to the previous Math Flasher program, this version permits negative numbers and decimal answers.

Deleting lines 120 and 130 from the easier Flasher program allows occasional negative numbers to result from subtraction problems and fractional (decimal) numbers to result from division.

Watch out! You may have to resort to the old-fashioned pencil-and-paper method of calculating answers to this program.

Program Listing

```
10 PRINT "ADDITION"
20 FOR L=1 TO 8
30 PRINT "**";
40 NEXT L
50 PRINT
60 PRINT
100 LET P=INT(100*RND)
110 LET Q=INT(100*RND)
200 PRINT "ADD ";P;" PLUS ";Q
210 INPUT R
220 CLS
300 IF R=P+Q THEN GOTO 400
310 PRINT "WRONG"
320 GOTO 410
400 PRINT "CORRECT"
410 PRINT P;" PLUS ";Q;" = ";P+Q
500 PRINT
510 PRINT
520 GOTO 10
```

Program Listing

```
10 PRINT "SUBTRACTION"
20 FOR L=1 TO 11
30 PRINT "**";
40 NEXT L
50 PRINT
60 PRINT
100 LET P=INT(100*RND)
110 LET Q=INT(100*RND)
200 PRINT "SUBTRACT ";Q;" FROM ";P
210 INPUT R
220 CLS
300 IF R=P-Q THEN GOTO 400
310 PRINT "WRONG"
320 GOTO 410
400 PRINT "CORRECT"
410 PRINT P;" MINUS ";Q;" = ";P-Q
500 PRINT
```

```
510 PRINT
520 GOTO 10
```

Program Listing

```
10 PRINT "MULTIPLICATION"
20 FOR L=1 TO 14
30 PRINT "*";
40 NEXT L
50 PRINT
60 PRINT
100 LET P=INT(100*RND)
110 LET Q=INT(100*RND)
200 PRINT "MULTIPLY ";P;" TIMES ";Q
210 INPUT R
220 CLS
300 IF R=P*Q THEN GOTO 400
310 PRINT "WRONG"
320 GOTO 410
400 PRINT "CORRECT"
410 PRINT P;" TIMES ";Q;" = ";P*Q
500 PRINT
510 PRINT
520 GOTO 10
```

Program Listing

```
10 PRINT "DIVISION"
20 FOR L=1 TO 8
30 PRINT "*";
40 NEXT L
50 PRINT
60 PRINT
100 LET P=INT(100*RND)
110 LET Q=INT(100*RND)
120 IF Q=0 THEN GOTO 110
200 PRINT "DIVIDE ";P;" BY ";Q
210 INPUT R
220 CLS
300 IF R=P/Q THEN GOTO 400
```



```
310 PRINT "WRONG"  
320 GOTO 410  
400 PRINT "CORRECT"  
410 PRINT P;" DIVIDED BY ";Q;" = ";P/Q  
500 PRINT  
510 PRINT  
520 GOTO 10
```

Exam Score Sorting

The final number scores of a large number of test results can be categorized and thereby cut down into a smaller quantity of numbers easily.

This program accepts exam scores and divides them into ranges we have labeled A, B, C, D and F. The program looks for test scores in a range of zero to 100. The predetermined grade ranges are F=0 to 59; D=60 to 69; C=70 to 79; B=80 to 89; and A=90 to 100.

You key in the letter X to break the entry cycle. Lines 100 to 140 sort the scores into letter grades A through F. Lines 150 to 170 sort highest and lowest scores.

Program Listing

```
10 GOSUB 400
20 PRINT "SCORE="
30 INPUT G$
40 CLS
50 IF G$="X" THEN GOTO 200
60 LET G=VAL G$
70 LET N=N+1
100 IF G<60 THEN LET F=F+1
110 IF G>59 AND G<70 THEN LET D=D+1
120 IF G>69 AND G<80 THEN LET C=C+1
130 IF G>79 AND G<90 THEN LET B=B+1
140 IF G>89 THEN LET A=A+1
150 IF N=1 THEN LET L=G
155 IF N=1 THEN LET H=G
160 IF G<L THEN LET L=G
170 IF G>H THEN LET H=G
180 GOTO 20
200 PRINT "TOTAL ";N;" SCORES"
210 PRINT "FROM ";L;" TO ";H
220 PRINT "A",A
230 PRINT "B",B
240 PRINT "C",C
250 PRINT "D",D
260 PRINT "F",F
270 STOP
```

```
400 CLEAR
410 LET A=Ø
420 LET B=Ø
430 LET C=Ø
440 LET D=Ø
450 LET F=Ø
460 LET N=Ø
470 RETURN
```

Astronomy

Lightyears/Distance Conversions

*Starlight,
Starbright,
I wish I may,
I wish I might,
Know the distance
To your light.*

For students of astronomy everywhere, here's how to plug your computer into your hobby: use the machine to discover distances across the Universe!

This program converts lightyears to kilometers or kilometers to lightyears or lightyears to miles or miles to lightyears. It's hard to visualize distances in lightyears. Run this program and you'll be better able to grasp the vast expanse of the Cosmos with your mind.

Of course, all distances are approximate. We use 365.86 days per year and, thus, 9.4830912×10^{12} km/ly or $5.892792872 \times 10^{12}$ mi/ly.

Program Listing

```
10 CLEAR
20 LET M=0
30 LET K=0
40 PRINT "CONVERSIONS"
50 FOR P=1 TO 11
60 PRINT "*";
70 NEXT P
80 PRINT
90 PRINT "1/LTYRS TO MILES"
100 PRINT "2/KM TO LTYRS"
110 PRINT "3/MILES TO LTYRS"
120 LET C$=INKEY$
130 IF C$="" THEN GOTO 120
140 IF C$>"1" THEN GOTO 400
300 CLS
310 PRINT "LTYRS: ";
```

```

320 INPUT L
330 PRINT L
340 LET K=L*(9.4830912*10**12)
350 LET M=L*(5.892792872*10**12)
360 PRINT "KM: ";K
370 PRINT "MILES: ";M
380 GOTO 550
400 CLS
410 IF C$="3" THEN GOTO 500
420 PRINT "KM: ";
430 INPUT K
440 PRINT K
450 LET L=K/(9.4830912*(10**12))
460 GOTO 540
500 PRINT "MILES: ";
510 INPUT M
520 PRINT M
530 LET L=M/(5.892792872*(10**12))
540 PRINT "LTYRS: ";L
550 PRINT
560 GOTO 10

```

Sample Run

CONVERSIONS

1/LTYRS TO MILES

2/KM TO LTYRS

3/MILES TO LTYRS

1

LTYRS:

4 ENTER

KM: 3.7932365 E+13

MILES: 2.3571172 E+13

CONVERSIONS

1/LTYRS TO MILES
2/KM TO LTYRS
3/MILES TO LTYRS

2

KM:
150000 ENTER
LTYRS: 1.5817627 E-8

CONVERSIONS

1/LTYRS TO MILES
2/KM TO LTYRS
3/MILES TO LTYRS

3

MILES:
150000 ENTER
LTYRS: 2.5454823 E-8

State Geographic Centers

This mind bender tests your knowledge of geographic locations of cities and towns in the United States. These are special places since, in each case, they are the town nearest to the geographic center of its state.

In other words, Columbus happens to be almost exactly in the center of Ohio. But which state has Challis at its center? Or Lewistown? Or Oklahoma City? (Well, some may be obvious!)

You not only learn a lot from running this program but you have a barrel of fun. Talk about trivia!

You can change the names. Here's a variety we'll bet you haven't seen anywhere else.

CLANTON, ALABAMA
MT. MCKINLEY, ALASKA
PRESCOTT, ARIZONA
LITTLE ROCK, ARKANSAS
MADERA, CALIFORNIA
PIKES PEAK, COLORADO
EAST BERLIN, CONNECTICUT
DOVER, DELAWARE
BROOKSVILLE, FLORIDA
MACON, GEORGIA
MAUI ISLAND, HAWAII
CHALLIS, IDAHO
SPRINGFIELD, ILLINOIS
INDIANAPOLIS, INDIANA
AMES, IOWA
CREAT BEND, KANSAS
LEBANON, KENTUCKY
MARKSVILLE, LOUISIANA
DOVER-FOXCROFT, MAINE
DAVIDSONVILLE, MARYLAND
WORCESTER, MASSACHUSETTS
CADILLAC, MICHIGAN
BRAINERD, MINNESOTA
CARTHAGE, MISSISSIPPI
JEFFERSON CITY, MISSOURI
LEWISTOWN, MONTANA

BROKEN BOW, NEBRASKA
AUSTIN, NEVADA
ASHLAND, NEW HAMPSHIRE
TRENTON, NEW JERSEY
WILLARD, NEW MEXICO
ONEIDA, NEW YORK
SANFORD, NORTH CAROLINA
MCCLUSKY, NORTH DAKOTA
COLUMBUS, OHIO
OKLAHOMA CITY, OKLAHOMA
PRINEVILLE, OREGON
BELLEFONTE, PENNSYLVANIA
CROMPTON, RHODE ISLAND
COLUMBIA, SOUTH CAROLINA
PIERRE, SOUTH DAKOTA
MURFREESBORO, TENNESSEE
BRADY, TEXAS
MANTI, UTAH
ROXBURY, VERMONT
BUCKINGHAM, VIRGINIA
WENATCHEE, WASHINGTON
SUTTON, WEST VIRGINIA
MARSHFIELD, WISCONSIN
LANDER, WYOMING

Program Listing

```
10 RAND
20 DIM S$(5,7)
30 DIM G$(5,8)
100 LET S$(1)="ALABAMA"
110 LET G$(1)="CLANTON"
120 LET S$(2)="IOWA"
130 LET G$(2)="AMES"
140 LET S$(3)="NEVADA"
150 LET G$(3)="AUSTIN"
160 LET S$(4)="TEXAS"
170 LET G$(4)="BRADY"
180 LET S$(5)="UTAH"
190 LET G$(5)="MANTI"
200 LET R=INT(6*RND)
210 IF R<1 THEN GOTO 200
```



```
300 PRINT "WHAT STATE HAS ITS"  
310 PRINT "CENTER NEAR ";G$(R)  
320 INPUT A$  
330 LET L=LEN A$  
340 CLS  
350 IF A$=S$(R,1 TO L) THEN GOTO 500  
400 PRINT "WRONG"  
410 GOTO 510  
500 PRINT "CORRECT"  
510 PRINT "STATE: ";S$(R)  
520 PRINT "CENTER: ";G$(R)  
530 PRINT  
540 CLEAR  
550 GOTO 10
```

Volumes

Cones. Cubes. Cylinders. Prisms. Pyramids. Spheres. Name your object. This program computes the volume and displays it in cubic units.

Put in inches, get cubic inches. Put in feet, get cubic feet. Yards, get cubic yards. No mixing units in any one calculation. Cylinder is right circular.

Program Listing

```
10 CLEAR
100 PRINT "OBJECT: ";
110 INPUT X$
120 PRINT X$
130 IF X$="CONE" OR X$="PYRAMID" OR X$=
    "PRISM" THEN GOTO 200
140 IF X$="CUBE" THEN GOTO 300
150 IF X$="CYLINDER" THEN GOTO 400
160 IF X$="SPHERE" THEN GOTO 500
170 GOTO 100
200 PRINT "AREA: ";
210 INPUT A
220 PRINT A
230 PRINT "HEIGHT: ";
240 INPUT H
250 PRINT H
260 LET V=A*H
270 IF X$="CONE" OR X$="PYRAMID" THEN
    LET V=V/3
280 GOTO 600
300 PRINT "LENGTH: ";
305 INPUT L
310 PRINT L
315 PRINT "WIDTH: ";
320 INPUT W
325 PRINT W
330 PRINT "HEIGHT: ";
335 INPUT H
340 PRINT H
345 LET V=L*W*H
```

```

350 GOTO 600
400 PRINT "RADIUS: ";
410 INPUT R
420 PRINT R
430 PRINT "HEIGHT: ";
440 INPUT H
450 PRINT H
460 LET V=2*PI*R*H
470 GOTO 600
500 PRINT "RADIUS: ";
510 INPUT R
520 PRINT R
530 LET V=(4*PI*(R**3))/3
600 PRINT "VOLUME: ";V
610 PRINT
620 PRINT
630 PRINT "PRESS A KEY"
640 IF INKEY$="" THEN GOTO 640
650 CLS
660 GOTO 10

```

Sample Run

```

OBJECT:
CONE ENTER
AREA
55 ENTER
HEIGHT
66 ENTER
VOLUME 1210

```

```

ENTER
OBJECT:
CYLINDER ENTER
RADIUS
29 ENTER
HEIGHT
15 ENTER
VOLUME 2733.1856

```

ENTER
OBJECT:
SPHERE ENTER
RADIUS
23 ENTER
VOLUME 50965.01

ENTER
OBJECT:
PRISM ENTER
AREA
71 ENTER
HEIGHT
18 ENTER
VOLUME 1278

ENTER
OBJECT:
CUBE ENTER
LENGTH
13 ENTER
WIDTH
8 ENTER
HEIGHT
24 ENTER
VOLUME 2496

Event Timer

Place your computer in a corner and let it time your next chess match. Three-minute egg. Final exam.

The computer asks how many minutes you want for the event you are timing, and then it notes when the time has passed.

You can calibrate the clock by changing the value of `P` in line 10. A larger number will slow down the clock. A smaller value for `P` will speed up the clock. As you can see we have started with a `P` value of 16. That works just about right on our TIMEX T/S 1000, but it is not necessarily right for your machine.

If you want to time an event of less than one minute, use a decimal. For instance, when you want to time a 30-second event, respond to the computer's inquiry with .5 or for 45 seconds key in .75. Use .17 for 10 seconds; .25 for 15 seconds.

Program Listing

```
10 LET P=16
20 LET C=0
30 PRINT "EVENT TIMER"
40 FOR L=1 TO 11
50 PRINT "*";
60 NEXT L
70 PRINT
80 PRINT "HOW MANY MINUTES?"
90 INPUT T
100 CLS
110 PRINT "PRESS ANY KEY TO START TIMING"
120 IF INKEY$="" THEN GOTO 120
130 CLS
140 LET C=C+1
150 IF C>60*T THEN GOTO 300
160 LET M=INT(C/60)
170 LET S=INT(C-(60*M))
180 PRINT M;" MINUTES ";S;" SECONDS"
190 FOR L=1 TO P
200 NEXT L
210 GOTO 130
```

```
300 CLS
310 PRINT "TIME IS UP"
320 PRINT T;" MINUTES HAVE PASSED"
330 FOR L=1 TO 8
340 PRINT
350 NEXT L
360 PRINT "TO TIME AGAIN, PRESS ANY KEY"
370 IF INKEY$="" THEN GOTO 370
380 CLEAR
390 GOTO 10
```

Class Roll Sorter

Here's a simple sorting routine which you can use to keep your class roll in order. Suppose it's the first day of classes and you have been handed an unorganized list of student names. Merely key those names into the computer and it will use this "bubble sort" program to put the list in alphabetical order. As set up, it accepts up to 20 names.

Program Listing

```
10 CLEAR
20 DIM M$(20,6)
30 PRINT "NAME?"
40 FOR L=1 TO 20
50 INPUT M$(L)
60 IF M$(L)(1)="X" THEN GOTO 80
70 NEXT L
80 CLS
90 LET T=0
100 FOR L=1 TO 19
110 PRINT "SORTING"
120 IF M$(L)<=M$(L+1) THEN GOTO 170
130 LET E$=M$(L)
140 LET M$(L)=M$(L+1)
150 LET M$(L+1)=E$
160 LET T=1
170 CLS
180 NEXT L
190 IF T=1 THEN GOTO 90
200 CLS
210 PRINT "CLASS ROLL:"
220 FOR L=1 TO 20
230 PRINT M$(L)
240 NEXT L
```


Programs for the business person

Marketing Performance

Ever stared at a page of numbers for hours without really seeing their relationships? Well, that doesn't have to happen when you put those numbers on a visual display—a graph.

This program is set up to take in data on each of 8 quarters and then display that data on a graph on your computer's video display.

The graph has 22 available positions from left to right. We have labeled those as $x1000$. You could make it dollars or millions or whatever you need.

Anyway you cut it, it's a short, easy-to-type-in program which can give you a quick look at how marketing performance in one quarter-year relates to marketing performance in other quarter years.

Program Listing

```
10 CLEAR
20 DIM N(8)
100 FOR P=1 TO 8
110 PRINT "DATA FOR QUARTER NUMBER "
    ;P; " ?"
120 INPUT N(P)
130 IF N(P)>22 THEN GOTO 120
140 CLS
150 NEXT P
200 FOR P=1 TO 8
210 PRINT "QTR " ;P;
220 FOR L=0 TO N(P)-1
230 PRINT CHR$ 128;
240 NEXT L
250 PRINT
260 PRINT
270 NEXT P
280 PRINT
290 PRINT TAB 10;"X 1000"
300 IF INKEY$="" THEN GOTO 300
310 CLS
320 GOTO 10
```

Daily Codes

Businesses everywhere are concerned about security. Banks, credit managers, warehousemen, shipping clerks, office managers, retail storekeepers, all need private daily codes for internal use to prevent unauthorized admission to private files, storage areas, financial records.

Now you can use your computer to generate a weekly set of codes for each day. If you feel insecure about a week's list in use, the computer will give you a new set of code numbers in a flash.

The computer generates a table of randomly-selected codes for seven days at a time.

Program Listing

```
10 GOSUB 300
100 PRINT "SUNDAY:      ";C
110 GOSUB 300
120 PRINT "MONDAY:      ";C
130 GOSUB 300
140 PRINT "TUESDAY:     ";C
150 GOSUB 300
160 PRINT "WEDNESDAY:   ";C
170 GOSUB 300
180 PRINT "THURSDAY:    ";C
190 GOSUB 300
200 PRINT "FRIDAY:      ";C
210 GOSUB 300
220 PRINT "SATURDAY:    ";C
230 STOP
300 LET C=INT(10000*RND)
310 IF C<1000 THEN GOTO 300
320 RETURN
```

Invoice Computer

There's a lot of repetitious math work to be done before you mail invoices to your customers. This software has the computer collect a few pertinent bits of data from you and then present all the various totals you need to plug into an invoice.

It gives you a total retail price for all goods sold on the invoice, total sales tax if applicable, shipping charges and the grand total amount due you from your customer.

Program Listing

```
10 PRINT "QUANTITY SOLD: ";
20 INPUT Q
30 PRINT Q
40 PRINT "UNIT PRICE: $";
50 INPUT P
60 PRINT P
70 PRINT "SALES TAX RATE PERCENT: ";
80 INPUT S
90 PRINT S
100 PRINT "SHIPPING CHARGES: $";
110 INPUT H
120 PRINT H
130 LET S=S*.01
140 LET C=Q*P
150 LET T=C*S
160 LET F=C+T+H
170 LET A=INT(100*C+.5)/100
180 LET B=INT(100*T+.5)/100
190 LET D=INT(100*F+.5)/100
200 CLS
210 PRINT "TOTAL PRICE = $";A
220 PRINT "SALES TAX = $";B
230 PRINT "SHIPPING CHARGES = $";H
240 PRINT
250 PRINT "INVOICE TOTAL = $";D
260 IF INKEY$="" THEN GOTO 260
270 CLS
280 GOTO 10
```

Sample Run

QUANTITY SOLD

178

UNIT PRICE = \$

55.98

SALES TAX RATE PERCENT =

6

SHIPPING CHARGES = \$

100

TOTAL PRICE = \$9964.44

SALES TAX = \$597.87

SHIPPING CHARGES = \$100

INVOICE TOTAL = \$10662.31

Hourly Wages

These useful lines compute total hours worked at regular pay and number of hours worked at time-and-a-half overtime. The computer then finds gross pay and rounds off to the nearest cent.

The program knows that overtime starts after 40 hours. It makes payroll bookkeeping quick and simple.

Program Listing

```
10 LET T=0
20 LET W=0
30 PRINT "HOURLY PAY RATE = $";
40 INPUT P
50 PRINT P
60 PRINT "NUMBER HOURS WORKED = ";
70 INPUT H
80 PRINT H
90 IF H>40 THEN LET T=H-40
100 IF H>40 THEN GOTO 140
110 LET W=H*P
120 PRINT "GROSS WAGES = $";W
130 STOP
140 LET W=(40*P)+(T*P*1.5)
150 GOTO 120
```

Sample Run

```
HOURLY PAY RATE = $
5.75
NUMBER HOURS WORKED =
61
GROSS WAGES = $411.125
```

Ad Campaign Profit

The ad salesman is standing in your office pressing for your answer. Do you want to advertise or not? Advertising costs plenty of money today. How can you make a quick decision about whether or not sales from advertising would be worth the cost?

In this program, the computer asks you for information about the list price of the item you would sell through advertising. It asks for the manufacturing cost of that item; the cost of the advertising campaign; and the number of units sold.

It computes your gross sales and deducts the cost of manufacturing and advertising to show an estimate of profits to be expected. If you key in a zero in response to the number-sold question, the machine will inquire as to the amount of profit you would like to make and then tell you how many units you would have to sell to make such a profit.

If you would like to make your own changes to this program listing, you'll want to know that memory location B holds the list price of the item you are selling; C is the unit manufacturing cost of the item; D is the total advertising cost; A is the number of units sold of the item; E is the profit on the sales of the item; and F is the profit you say you want to make.

Suppose you have an item you sell for \$9.95 and it costs you \$1.25 to produce it. An ad campaign costing \$330 results in sales of 50 units. Your profit from the campaign would be \$105. If you only wanted to know how many units you would have to sell to break even, enter zero in response to the number-sold question and \$1 to the profit-wanted question. You'll discover you need to sell just over 38 units to break even.

Program Listing

```
10 PRINT "ITEM LIST PRICE: ";  
20 INPUT B  
30 PRINT "$";B  
40 PRINT "MFG COST: ";
```



```

50 INPUT C
60 PRINT "$";C
70 PRINT "AD COST: ";
80 INPUT D
90 PRINT "$";D
100 PRINT "QUANTITY SOLD: ";
110 INPUT A
120 PRINT A
130 IF A=Ø THEN GOTO 300
200 LET E=A*B-A*C-D
210 PRINT "$";E;" PROFIT"
220 GOTO 400
300 PRINT "PROFIT WANTED: $";
310 INPUT F
320 PRINT F
330 LET A=(F+D)/(B-C)
340 PRINT "YOU MUST SELL ";A
350 PRINT "FOR $";F;" PROFIT"
400 PRINT
410 PRINT
420 PRINT "TO DO ANOTHER, PRESS ANY KEY"
430 IF INKEY$="" THEN GOTO 430
440 CLS
450 GOTO 10

```

Sample Run

```

ITEM LIST PRICE: $9.95
MFG COST: $1.25
AD COST: $330
QUANTITY SOLD: 50
$105 PROFIT

```

TO DO ANOTHER, PRESS ANY KEY

```

ITEM LIST PRICE: $9.95
MFG COST: $1.25
AD COST: $330
QUANTITY SOLD: Ø
PROFIT WANTED: $1
YOU MUST SELL 38.045977
FOR $1 PROFIT

```

Media Money Massage

If you have used the *Ad Campaign Profit* program earlier in this book, you know how many bucks you can expect to make from advertising. But, suppose two salesmen are standing in your office. One from your local newspaper and the other from a local television station. Both want your advertising dollar and you can't decide which is the best buy. Let your computer decide!

This program compares the cost of advertising in two media and reports which is most favorable. First it computes cost-per-thousand. Then it highlights the least-expensive medium.

Questions are asked by the computer display panel. The report compares any media—newspaper, radio, television, magazines, shoppers, etc.

Imagine your friendly salesmen are from the Daily Post and the Evening News. The ad in the Post costs \$250. In the News it is \$300. The Post's circulation is 27,500 readers. The News has 32,500 readers. Which is the better buy? The Post is about 14¢ cheaper per thousand readers.

Program Listing

```
10 PRINT "FIRST MEDIUM: ";
20 INPUT N$
30 PRINT N$
40 PRINT "AD COST: $";
50 INPUT A
60 PRINT A
70 PRINT "CIRCULATION: ";
80 INPUT C
90 PRINT C
100 LET M=1000*(A/C)
110 PRINT "SECOND MEDIUM: ";
120 INPUT P$
130 PRINT P$
140 PRINT "AD COST: $";
150 INPUT Q
```

```

160 PRINT Q
170 PRINT "CIRCULATION: ";
180 INPUT R
190 PRINT R
200 LET S=1000*(Q/R)
210 PRINT
220 PRINT
230 PRINT N$;" CPM: $";M
240 PRINT P$;" CPM: $";S
250 IF S>M THEN PRINT N$;" IS LOWER"
260 IF M>S THEN PRINT P$;" IS LOWER"
270 IF M=S THEN PRINT "NO DIFFERENCE"
300 PRINT
310 PRINT
320 PRINT "TO DO MORE, PRESS ANY KEY"
330 IF INKEY$="" THEN GOTO 330
340 CLS
350 GOTO 10

```

Sample Run

FIRST MEDIUM:
POST ENTER

AD COST: \$
250 ENTER

CIRCULATION:
27500 ENTER

SECOND MEDIUM:
NEWS ENTER

AD COST: \$
300 ENTER

CIRCULATION:
32500 ENTER

POST CPM: \$9.0909091
NEWS CPM: \$9.2307692
POST IS LOWER

TO DO MORE, PRESS ANY KEY

Salesman's Commission

Representatives, salesmen, account representatives, sales representatives. Here's the no-sweat way to compute commissions to be paid to your sales corps.

The computer will ask you for pertinent data and then display results including the salesman's name, the pay period, his commission percentage rate, gross sales, and commission payable.

Program Listing

```
10 PRINT "SALESMAN COMMISSION"
20 FOR L=1 TO 19
30 PRINT "**";
40 NEXT L
50 PRINT
100 PRINT "PERIOD ENDING DATE: ";
110 INPUT D$
120 PRINT D$
200 PRINT "SALESMAN NAME: ";
210 INPUT N$
220 PRINT N$
300 PRINT "COMMISSION PERCENT: ";
310 INPUT P
320 PRINT P
330 LET K=P*.01
400 PRINT "GROSS SALES: ";
410 INPUT Q
420 PRINT "$";Q
430 LET T=K*Q
500 PRINT "COMMISSION: $";T
510 PRINT
520 PRINT
530 PRINT "FOR ANOTHER, PRESS ANY KEY"
600 IF INKEY$="" THEN GOTO 600
610 CLS
620 GOTO 10
```

Sample Run

SALESMAN COMMISSION

PERIOD ENDING DATE:

12/31/84 ENTER

SALESMAN NAME:

SMITH ENTER

COMMISSION PERCENT:

15 ENTER

GROSS SALES: \$

16243.00 ENTER

COMMISSION: \$2436.45

FOR ANOTHER, PRESS ANY KEY

Unit Price

Suppose you find 895 green Widgets and buy them for \$695. How much did each green Widget cost? Rounded off, \$.77.

Unit price is total price divided by quantity. The quantity can be expressed in weight, total numbers, etc. It works the same whether you are talking about pounds of coffee, yards of concrete, gallons of ice cream, boxes of books, or units of Widgets.

This program asks for the name of the item, quantity purchased and total price paid. It then displays quantity, name, total and unit price.

Program Listing

```
10 LET U=0
20 PRINT "ITEM NAME IS ";
30 INPUT N$
40 PRINT N$
50 PRINT "QUANTITY OF ITEMS = ";
60 INPUT Q
70 PRINT Q
80 PRINT "TOTAL PRICE PAID FOR ITEMS
   = $";
90 INPUT P
100 PRINT P
110 LET U=P/Q
120 PRINT N$;" UNIT PRICE = $";U
200 PRINT
210 PRINT
220 PRINT "TO DO ANOTHER, PRESS ANY KEY"
230 IF INKEY$="" THEN GOTO 230
240 CLS
250 GOTO 10
```

Sample Run

ITEM NAME IS
WIDGETS
QUANTITY OF ITEMS =
999
TOTAL PRICE PAID FOR ITEMS = \$
14653

BEEP WIDGETS UNIT PRICE = \$14.667668

Executive Decision Maker

This is handy for the busy executive who doesn't have time for decisions.

Line 10 clears the screen. Line 20 generates a random number from zero to 99. Line 30 selects a yes answer if the random number is greater than 49. Otherwise, line 40 chooses a *no* answer.

Program Listing

```
10 CLS
20 LET X=100*RND
30 IF X>49 THEN PRINT "YES"
40 IF X<49 THEN PRINT "NO"
50 IF INKEY$="" THEN GOTO 50
60 GOTO 10
```

In this superior edition, a choice of eight replies is possible.

Program Listing

```
10 DIM D$(8,16)
20 RAND
100 LET D$(1)="YES"
110 LET D$(2)="FIRE SOMEONE"
120 LET D$(3)="PASS THE BUCK"
130 LET D$(4)="MAYBE"
140 LET D$(5)="REORGANIZE"
150 LET D$(6)="NO"
160 LET D$(7)="SEE YOUR ANALYST"
170 LET D$(8)="SIT ON IT"
200 LET R=INT(9*RND)
210 IF R<1 THEN GOTO 200
300 PRINT D$(R)
400 IF INKEY$="" THEN GOTO 400
410 CLS
420 GOTO 10
```


Cash Receipts Comparer

Business been up? Or, dropping off? How have your cash receipts looked over the last six months?

This short, easy-to-key-in piece of software accepts data from you about each month's receipts and then displays that data in an easy-to-read graph. The graph shows exactly how one month's receipts compare with another.

Lines 200 to 290 create the graph.

By the way, so the graph won't overflow the screen, The graph has 22 available positions from left to right. You could make it dollars or millions or whatever you need.

Anyway you cut it, it's a short, easy-to-type-in program which can give you a quick look at how marketing performance *total receipts* in one *month* relates to marketing performance in other months.

Program Listing

```
10 CLEAR
20 DIM R(6)
100 FOR M=1 TO 6
110 PRINT "DATA FOR MONTH ";M;" ?"
120 INPUT R(M)
130 IF R(M)>22 THEN GOTO 120
140 CLS
150 NEXT M
200 FOR M=1 TO 6
210 PRINT "MONTH ";M;
220 FOR L=0 TO R(M)-1
230 PRINT CHR$ 128;
240 NEXT L
250 PRINT
260 PRINT
270 NEXT M
280 PRINT
290 PRINT TAB 10;"RECEIPTS"
300 IF INKEY$="" THEN GOTO 300
310 CLS
320 GOTO 10
```

Mark Up

Mr. Storekeeper, here's just what you have needed to compute mark ups. This program causes your TIMEX to find the retail price for which your percentage off would give the wholesale cost.

For instance, if you got 40 percent off on an item and paid \$60, how much was it priced at, at retail? The answer is \$100. To put that another way, if retail price or suggested retail price is \$100 and you got 40 percent off at wholesale, what is the wholesale price? The answer is \$60.

Try \$40 wholesale which is 60 percent off. The answer is \$100 retail. Or try \$10 wholesale at 90 percent off. Retail would be \$100. Or \$75 wholesale at 25 percent off gives \$100 retail.

Here's a toughie! Try \$19.95 wholesale cost. Mark-up percentage is 40. The correct retail answer is \$33.25.

Program Listing

```
10 LET D=0
20 LET R=0
30 PRINT "WHOLESALE COST = $";
40 INPUT W
50 PRINT W
60 PRINT "MARK-UP PERCENTAGE = ";
70 INPUT P
80 PRINT P
90 CLS
100 LET D=1-.01*P
110 LET R=W/D
120 PRINT "RETAIL PRICE = $";R
130 PRINT
140 PRINT
150 PRINT "TO DO MORE, PRESS ANY KEY"
160 IF INKEY$="" THEN GOTO 160
170 CLS
180 GOTO 10
```

Appendix

Appendix A: Sinclair BASIC Words

Here is a convenient list, with short explanations, of each of the BASIC language words as used by Sinclair in the ZX-81 and in the TIMEX/Sinclair 1000 computer:

Functions

ABS	absolute value of a number
ACS	arccosine
AND	logical AND
ASN	arcsine
ATN	arctangent
CHR\$	changes 0 to 255 number to a character
CODE	number of first character in a string
COS	cosine
EXP	exponent
INKEY\$	scans keyboard
INT	integer part of number
LEN	length of a string
LN	natural logarithm
NOT	logical NOT
OR	logical OR
PEEK	look at one memory-address location
PI	3.14159265
RND	random-number generator
SGN	sign of a number
SIN	sine
SQR	square root
STR\$	change numerical value to string
TAN	tangent
USR	call machine-language routine
VAL	change string to number
+	addition of number or strings
-	subtraction
*	multiplication
/	division
**	raising to a power
=	equals
>	greater than
<	less than
<=	less than or equal to
>=	greater than or equal to
<>	not equal to

Statements

CLEAR	clears variables
CLS	clears screen
CONT	continue after STOP
COPY	copy TV screen on printer
DIM	dimensions numerical array
DIM \$	dimensions string array
FAST	starts fast mode of TV display
FOR	first part of FOR/NEXT loop
GOSUB	jump to subroutine
GOTO	jump to a line
IF	first part of IF/THEN decision maker
INPUT	stops so data can be entered
LET	assigns a value to a variable
LIST	displays program list on TV screen
LLIST	displays program list on printer
LOAD	transfers program from tape into computer
LPRINT	PRINT on printer
NEW	erases all of program memory and variables
NEXT	last part of FOR/NEXT loop
PAUSE	holds TV display on for brief delay
PLOT	blacks in a graphics-screen dot
POKE	writes in one memory-address location
PRINT	displays on TV screen
RAND	reseeds random-number generator
REM	statement ignored during run
RETURN	end of GOSUB, jump back to main routine
RUN	clears variables and starts program action
SAVE	records program from computer onto tape
SCROLL	rolls TV display up a line
SLOW	switches to slower display mode
STEP	optional part of FOR/NEXT loop
STOP	temporary halt in a run
TAB	PRINT at a certain place on TV display
THEN	last part of IF/THEN decision maker
TO	follows FOR in FOR/NEXT loop
UNPLOT	blanks out a graphics-screen dot

Appendix B: Error Messages

This is a list of what Sinclair calls *Report Codes* and what you may refer to as error messages. These are notes to you from the computer about mistakes you may have made and about the status of the computer:

Code Number/ Letter	Description of Error or Status
0	successful completion
1	you have a NEXT without a matching FOR
2	you forgot to use LET or DIM or your FOR variable wrong
3	a subscript is wrong
4	there's not enough memory space for what you want to do
5	the TV screen is full
6	you have computed too large a number
7	you have a RETURN without a matching GOSUB
8	INPUT can only be used in a program line
9	CONT can't restart at STOP
A	you are not using the function correctly
B	an integer is out of range
C	you can't VAL that string
D	you used BREAK to interrupt a run or you are using STOP in an INPUT line
F	there is no such program name

books from ARCsoft Publishers

For the TRS-80 PC-1, PC-2, Sharp PC-1211, PC-1500:

99 Tips & Tricks for the New Pocket Computers (PC-2/PC-1500)		
128 pages	\$7.95	ISBN 0-86668-019-5
101 Pocket Computer Programming Tips & Tricks		
128 pages	\$7.95	ISBN 0-86668-004-7
Pocket Computer Programming Made Easy		
128 pages	\$8.95	ISBN 0-86668-009-8
Murder In The Mansion and Other Computer Adventures		
96 pages	\$6.95	ISBN 0-86668-501-4
50 Programs in BASIC for Home, School & Office		
96 pages	\$9.95	ISBN 0-86668-502-2
50 MORE Programs in BASIC for Home, School & Office		
96 pages	\$9.95	ISBN 0-86668-003-9
Pocket-BASIC Coding Form programming worksheet tablets		
40-sheet pad	\$2.95	ISBN 0-86668-801-3

For the TRS-80 Color Computer:

101 Color Computer Programming Tips & Tricks		
128 pages	\$7.95	ISBN 0-86668-007-1
55 Color Computer Programs for Home, School & Office		
128 pages	\$9.95	ISBN 0-86668-005-5
55 MORE Color Computer Programs for Home, School & Office		
112 pages	\$9.95	ISBN 0-86668-008-X
Color Computer Graphics		
128 pages	\$9.95	ISBN 0-86668-012-8
The Color Computer Songbook		
96 pages	\$7.95	ISBN 0-86668-011-X
My Buttons Are Blue and Other Love Poems		
96 pages	\$4.95	ISBN 0-86668-013-6
Color Computer BASIC Coding Form program worksheet tablets		
40-sheet pad	\$2.95	ISBN 0-86668-802-1

For the APPLE Computer:

101 APPLE Computer Programming Tips & Tricks		
128 pages	\$8.95	ISBN 0-86668-015-2
33 New APPLE Computer Programs for Home, School & Office		
96 pages	\$8.95	ISBN 0-86668-016-0
APPLE Computer BASIC Coding Form program worksheet tablets		
40-sheet pad	\$2.95	ISBN 0-86668-803-X

For the ATARI 400 and 800 computers:

101 ATARI Computer Programming Tips & Tricks		
128 pages	\$8.95	ISBN 0-86668-022-5
31 New ATARI Computer Programs for Home, School & Office		
96 pages	\$8.95	ISBN 0-86668-018-7
ATARI Computer BASIC Coding Form program worksheet tablets		
40-sheet pad	\$2.95	ISBN 0-86668-806-4

books from ARCsoft Publishers

For the Sinclair ZX-81 and TIMEX/Sinclair 1000 computers:

101 TIMEX 1000 and Sinclair ZX-81 Programming Tips & Tricks		
128 pages	\$7.95	ISBN 0-86668-020-9
37 TIMEX 1000 & Sinclair ZX-81 Programs for Home, School & Office		
96 pages	\$8.95	ISBN 0-86668-021-7
TIMEX 1000 & Sinclair ZX-81 BASIC Coding Form program worksheets		
40-sheet pad	\$2.95	ISBN 0-86668-807-2

For the electronics hobbyist:

25 Quick-N-Easy Electronics Projects		
96 pages	\$4.95	ISBN 0-86668-023-3
25 Electronics Projects for Beginners		
96 pages	\$4.95	ISBN 0-86668-017-9
25 Easy-To-Build One-Night & Weekend Electronics Projects		
96 pages	\$4.95	ISBN 0-86668-010-1

Computer program-writing worksheets:

Each in tablet of 40 sheets with stiff backing

Universal BASIC Coding Form	\$2.95
IBM Personal Computer BASIC Coding Form	\$2.95
ATARI Computer BASIC Coding Form	\$2.95
Pocket Computer BASIC Coding Form	\$2.95
APPLE Computer BASIC Coding Form	\$2.95
Color Computer BASIC Coding Form	\$2.95
TIMEX 1000/Sinclair ZX-81 BASIC Form	\$2.95

ISBN: International Standard Book Number

ARCsoft Publishers

Post Office Box 132

Woodsboro, Maryland 21798

37 TIMEX 1000/Sinclair ZX-81 Programs for Home, School & Office

by Edward Page

For the home, classroom or office, an all-new volume of tested ready-to-type-and-run software in BASIC for the TIMEX/Sinclair 1000 and the Sinclair ZX-81 personal computers.

For the home, monthly loan payment, bowling score sorter, savings account interest computer, wisecracking insulter, even a program to turn your T/S 1000 or ZX-81 into a one-armed bandit.

For the student and teacher, foreign capitals, areas, volumes, photography, astronomy, math flasher, event timer, even a super exam-score sorter.

For the small-business person, daily code chart, profit estimator, invoice computer, payroll, ad profits, unit price, gross and net, even the world's fastest executive decision maker!

All of these and more are among the 37 different programs in this one handy volume. Each is complete and will run all by itself on your T/S 1000 or ZX-81. Or it can be blended into a larger set of program lines.

All of these programs have been designed for easy and quick typing into your TIMEX/Sinclair 1000 or Sinclair ZX-81 computer. These are ideal programs for newcomers, novices, beginners and first timers in the world of computers. By typing in these programs, you will see how software is written. You'll discover what to do, how to do it, when to make changes. You'll find hours of operating fun, practicality and usefulness.

This book includes a handy appendix of all words in the Sinclair version of the BASIC programming language. And a list of error messages/report codes used in the T/S 1000 and ZX-81. Even if you don't own a computer yet, reading these program lines will let you learn programming by following their logic.

These programs will run on other popular microcomputers, programmed in BASIC, with only minor modifications to program lines. With only slight changes, use these on TRS-80 Model I, II, III, 16, Color, Pocket, Commodore VIC, Max Machine, Pet, Model 64, Apple II, Sharp, Casio, Panasonic, Hewlett-Packard, Quasar, Atari 400/800, and others.

ARCsoft Publishers

WOODSBORO, MARYLAND

ISBN 0-86668-021-7